


SPECTRAVIDEO

THIS IS THE OFFICIAL NEWSLETTER FOR THE
SPECTRAVIDEO USERS GROUP OF TASMANIA



News Letter

CONTENTS	PAGE
<i>EDITOR'S COMMENT</i>	2
<i>DEMONSTRATION PROGRAM</i>	3
<i>DRAW COMMAND ERROR OR NOT???</i>	4
<i>ANOTHER PECULIARITY</i>	4
<i>COLECO ADAPTER REVIEW</i>	5
<i>THE MERGE COMMAND</i>	7
<i>BOUNCE</i>	9
<i>EXPLORING BASIC Pt-2</i>	10
<i>PROGRAM OF THE MONTH "ROBONS"</i>	13
<i>JESSUPS</i>	16
<i>ACHTUNG</i>	16

ISSUE No.
1 - 6
DATE
MAR - 1984

ALL CORRESPONDENCE TO:

*S.V.U.G.T.,
P.O. BOX 191,
SOUTH LAUNCESTON
TAS. 7250*

PHONE : (003) 312648

MEMBERSHIP FEES

AUSTRALIA \$15.00
OVERSEAS \$20.00
OVERSEAS AIRMAIL .. \$25.50

SPECTRAVIDEO

COMPUTER USERS GROUP OF TASMANIA

EDITOR'S COMMENT

Hi folks! Firstly I wish to thank Mr. M Gasiorowski, Mr. L.A. Dunning and Mr. J. Collins for their contributions to this newsletter, and to Mr B. Scott for the program of the month. Special thanks to Mr M.J. Tyeson for spending many hours hitting the computer keys. His help was much appreciated.

A few interesting events have occurred over the last month that I would like to pass on to you.

The most important is that the SPECTRAVIDEO distribution in Australia has changed hands. It is now being carried out by:

Rose Music Pty. Ltd.,
17-33 Market St,
South Melbourne,
Victoria. 3205 (03 699 2388)

On behalf of the club members I would like to welcome them to the SPECTRAVIDEO family.

Many of you members have requested that our users group be renamed, to include all Australia. We thought about it and as the majority of our members are mainlanders we have decided to change the name to SPECTRAVIDEO AUSTRALASIAN USERS GROUP (S.A.U.G.). This is only fair considering that in addition to all our Australian members we also have members from Papua New Guinea and New Zealand. For interest, below is what you will see at the top of each page as from the next issue.

SPECTRAVIDEO™
AUSTRALASIAN USERS GROUP

All of us are still hungry for more written information. I received an SV328 Manual in the mail some weeks ago. It's a big improvement on the SV318 Manual and worth asking your retailer about it. I have heard on the grapevine of a book written by Microsoft on the MSX BASIC. Please if anyone knows where I can get one write to me or better still send one, I will be happy to repay the costs. Also I would like to hear from members on BASIC books they find useful for their SV's. E.G. Manuals from other computers with similar or the same BASIC. Remember G.W. BASIC is very close to our own. If enough replies are received I intend to write an article in a future news letter.

MEMORY MAP We need one for machine code programming. I know from my sources that one is available but why can't we get it? Most other computer manufacturers are more than willing to supply theirs to the users. I suggest that you all write to SV in Hong Kong (There's power in numbers).

SPECTRAVIDEO

COMPUTER USERS GROUP OF TASMANIA

DEMONSTRATION PROGRAM

On page 107 of the SV318 Personal Computer User's Manual the following is written.

This program demonstrates the use of the joystick to move sprites. When run, a small spaceship-like sprite will appear. This sprite can be moved anywhere within the confines of the screen and will also fire a bullet when the trigger is pressed.

Well the User Manual was up to it's usual standard and the program won't work. However SPECTRAVIDEO has made amends in their SV328 Manual with a working copy of the program. So for all you SV318 Manual owners here is the working program.

```
10 COLOR 15,1,1
20 SCREEN 1,2
30 REM This section reads in the SPRITES
40 FOR T=1 TO 8
50 READ A$
60 S$=S$+CHR$(VAL("&B"+A$))
70 NEXT T
80 SPRITE$(1)=S$
90 FOR T=1 TO 8
100 READ B$
110 U$=U$+CHR$(VAL("&B"+B$))
120 NEXT T
130 SPRITE$(2)=U$
140 REM This section sets the initial location of the SPRITE
150 X=128 : Y=96
160 PUT SPRITE 1,(X,Y),9,1
170 D=STICK(0)
180 F=STRIG(0)
190 REM This section takes the given joystick information and uses it to
200 REM make the SPRITE move.
210 IF F<>0 THEN GOSUB 470
220 IF D=1 THEN X=X : Y=Y-1
230 IF D=2 THEN X=X+1 : Y=Y-1
240 IF D=3 THEN X=X+1 : Y=Y
250 IF D=4 THEN X=X+1 : Y=Y+1
260 IF D=5 THEN X=X : Y=Y+1
270 IF D=6 THEN X=X-1 : Y=Y+1
280 IF D=7 THEN X=X-1 : Y=Y
290 IF D=8 THEN X=X-1 : Y=Y-1
300 GOTO 160
310 DATA 00111100
320 DATA 01000010
330 DATA 10000001
340 DATA 11111111
350 DATA 01000010
360 DATA 10000001
370 DATA 10000001
380 DATA 10000001
390 DATA 00010000
400 DATA 00101000
410 DATA 00101000
420 DATA 00111000
430 DATA 00000000
440 DATA 00000000
450 DATA 00000000
460 DATA 00000000
470 FOR I=Y-3 TO -20 STEP -2
480 PUT SPRITE 0,(X,I),9,2
490 NEXT I
500 RETURN
```


SPECTRAVIDEO

COMPUTER USERS GROUP OF TASMANIA

DRAW COMMAND ERROR OR NOT???

by M. Gasiorowski

I would like to mention some oddities and some handy tips I have picked up. When using the DRAW command, I find that the 'move relative' command will not work as specified in the BASIC Reference Guide. To state an example:

```
DRAW"C14M127,95M+8,-20"    will not work
DRAW"C14M127,95M-8,+20"    will not work
```

But

```
DRAW"C14M127,95M+=B; ,=C;    will work
```

In both cases if B and C have the values 8,-20 in the first case and -8,20 in the second. Also:

```
DRAW"C14M127,95M+10,20"    will work
```

So, using negative values produces a function call error, whereas positive values or variables give no error and work correctly!

The BASIC Reference Guide, however, states that both negative and positive values should work:

'M(x,y) Move absolute or relative. If x is preceded by a "+" or "-", x and y are added to the current graphics position, and connected to the current graphics position, with a line. Otherwise, a line is drawn to point x,y from the current position'.

Does anyone out there know why this does not work as specified? (Or is there an ambiguity in the manual; or have I missed something?)

ANOTHER PECULIARITY

by M. Gasiorowski

Try this:

```
10 SCREEN,0
20 FOR X=0 TO 1000 : VPOKE X,191
30 NEXT X : LOCATE 0,24 : PRINT "SPECTRAVIDEO"
```

Type it in, and RUN it.

Then using the cursor keys or joystick cursor control, move the cursor over the letters "SPECTRAVIDEO". The VPOKED part of the screen will change to the same characters as the cursor!

Does anyone know of other peculiarities?

SPECTRAVIDEO

COMPUTER USERS GROUP OF TASMANIA



COLECO ADAPTER REVIEW

by J. Collins

Being an avid Computer Games fan and having seen some COLECOVISION games in various stores, I was quite pleased to note the availability of an adapter allowing use of COLECO Cartridges on SV318/328 systems.

After visiting my friendly Bank Manager I decided to take the plunge and placed an order for an SV603 GAME ADAPTER and received my first shock. Although it isn't mentioned anywhere in the Spectravideo literature you need to purchase at least one Quickshot III Joystick. The Quickshot III is required because once up and running the Coleco Adapter locks out your normal keyboard and without the keypad on the Quickshot III you cannot select the number of players or level of difficulty. As well as these facilities you need the 0-9 keys for some games which interact continuously with the keypad. Mousetrap is a good example of this requirement.

The unit is solidly presented in a neat, off white plastic case. The P.C.B. is of very high quality with layout and accessibility very good should service ever be required. There are 12 IC's and an 8k monitor program in an EPROM.

On the top of the case is the cartridge slot with a hinged flap. Cartridges click in easily and are held firmly. On the right hand side are two ports for Spectravideo type Joysticks and on the lower left front the 50 pin connector for connection to your SV318/328. This poses a small problem for those users who have an expander unit. In order to use the games adapter you must remove your expander and vice versa, although originally it was intended that the Coleco Game Adapter could be plugged in and then other peripherals would plug into the rear of it. All would function together as a normal computer until a games cartridge was plugged in. The 50 pin buss is available on the rear of the P.C.B. in the adapter but the necessary hole in the case and switching arrangement have not been utilised in the models on sale in Australia. I don't know why not but it may be interesting to find out. This would make the adapter much more useable and attractive to potential buyers if they could plug it in and leave it there.

How does it perform??? The best answer I can give is that it performs very well. I've tried several of the latest Coleco Cartridges and am very impressed. You get a sign on message and after a slight pause you are presented with the games level selection screen and from there you go direct to the start of the game at the level selected by the keypad choice.




ZAXXON

SPECTRAVIDEO

COMPUTER USERS GROUP OF TASMANIA

The Quickshot III Joystick is a "JOY" to work with. Response is very good and the keypad is easy to use where required. As stated earlier "Mousetrap" uses the keypad continuously during play requiring manipulation of 4 keys and the joystick. It's easier than it sounds and quite smooth and fast. My 10 year old son gives it his top vote as do all his friends. My adult friends agree wholeheartedly.




PEPPER II

Not cheap at around \$130.00 for the adapter and \$40.00 for Quickshot III but when viewed in balance with entertainment value and the superb quality of the Colecovision games I think it's a winner, especially as the advertised cartridge software for our own systems are still not available in Australia. If you get a chance to look at Space Panic, Zaxxon, and Mousetrap they're all superb. (I recommend Donkey Kong Jr., Pepper II and Turbo. ED.)

If permitted I will review cartridges as they come to hand.

Footnote from ED.

Please note Coleco expansion module #2 (as used with Turbo) will work with our Coleco adapter but expansion module #1 (for Atari Games) will not.



DONKEY KONG



SPECTRAVIDEO

COMPUTER USERS GROUP OF TASMANIA

THE MERGE COMMAND

by ed.

I have recieved quite a few letters requesting assistance with the MERGE command used in BASIC. (So here it is.)

The MERGE command may be used with cassette and disk systems and works in exactly the same way. I will give examples for both in this article.

First to explain the MERGE command : It is used to allow two programs to be MERGED (overlaid).

For Example if you MERGE:

```
10 PRINT "HELLO"  
30 END
```

with

```
5 CLS  
10 GOTO 20  
20 PRINT 55*8
```

you would get

```
5 CLS  
10 PRINT "HELLO"  
20 PRINT 55*8  
30 END
```

NOTE* How line 10 in the second listing (10 GOTO 20) is lost when MERGED with the first as the line 10 (10 PRINT "HELLO") will overlay the existing line 10.

Now how to do it:

Type in the first program

```
10 PRINT "HELLO"  
30 END
```

Now save it to Casette or Disk in ASCII format. This is a special format and is invoked by adding ",A" to a SAVE instruction.

```
SAVE "1:ONE",A          disk version  
SAVE "CAS:ONE",A       cassette version  
                        (NOTE* use SAVE not CSAVE)
```

Now type "NEW" for clear memory.

Type in the second program

SPECTRAVIDEO

COMPUTER USERS GROUP OF TASMANIA

```
5 CLS
10 GOTO 20
20 PRINT 55*8
```

Now if you use a cassette rewind to the beginning of the first program. You disk users dont need to do anything.

Now use the MERGE command to bring the two programs together.

```
MERGE "1:ONE"          disk version
```

```
MERGE "CAS:ONE"       Cassette version
```

Now type list and Hey Presto the two programs are one. The new program can now be saved as normal on cassette with CSAVE and disk with SAVE.



SPECTRAVIDEO

COMPUTER USERS GROUP OF TASMANIA

BOUNCE

by M. Gasiorowski

I have decided to send you one of my trivial programs that does nothing much but is quite interesting. If nothing else, it demonstrates the use of sprites, as well as parabolic motion of falling objects.

```
10 DEFSNG A-Z
20 COLOR 15,1 : SCREEN 0,0 : CLS : LOCATE 17,6 : PRINT "Bounce" : LOCATE ,15 : I
NPUT "Number of balls";NB
30 INPUT "Maximum speed";E
40 INPUT "Bounce Coefficient";AO
50 SCREEN 1,1 : COLOR 15,1,1 : CLS
60 LINE (9,176)-(252,176),4
70 BEEP : R=RND(-TIME)
80 LINE (252,0)-(252,176),4
90 LINE (9,0)-(9,176),4
100 FOR T=1 TO 8
110 READ A#
120 S#=S#+CHR$(VAL("&B"+A#))
130 NEXT T
140 SPRITE$(1)=S#
150 DATA 00011000
160 DATA 00100100
170 DATA 01000010
180 DATA 01011010
190 DATA 01011010
200 DATA 01000010
210 DATA 00100100
220 DATA 00011000
230 DIM X(NB),Y(NB),ZA(NB),AB(NB),XC(NB),AY(NB)
240 FOR B=1 TO NB : X(B)=127 : Y(B)=0 : ZA(B)=E-E*2*RND(1) : AB(B)=.8+.1*RND(1)
: XC(B)=2+13*RND(1)
250 IF AO>0 THEN AB(B)=AO
260 NEXT B
270 FOR Q=1 TO 300 : FOR B=1 TO NB
280 AY(B)=AY(B)+1 : Y(B)=Y(B)+AY(B) : X(B)=X(B)+ZA(B)
290 IF Y(B)>160 THEN A(B)=320-Y(B) : AY(B)=-AY(B)*AB(B) : ZA(B)=ZA(B)*.97 : IF A
BS(AY(B))<1.5 THEN AY(B)=0 : Y(B)=160 ELSE SOUND 8,8 : SOUND 8,0
300 IF X(B)>238 THEN X(B)=476-X(B) : ZA(B)=-ZA(B)*AB(B)
310 IF X(B)<9 THEN X(B)=18-X(B) : ZA(B)=-ZA(B)*AB(B)
320 PUT SPRITE B,(X(B),Y(B)),XC(B),1
330 NEXT B,Q : RUN
```

SPECTRAVIDEO

COMPUTER USERS GROUP OF TASMANIA

EXPLORING BASIC Pt-2

by L.A. Dunning

BASIC is no more than a machine code program. This rather self evident point is normally not apparent to beginners with microcomputers. As such, it can be divided into several different areas, each with certain functions. A typical BASIC interpreter might be composed of the following modules:

- System Boot
- Interpreter
- Editor
- Statement Code
- Functions
- Input/Output Drivers
- Keyboard Driver
- Vectors
- Buffers
- Tables

Let's take these one at a time. The System Boot is the section of code used when the system is initially turned on or reset. This will set up the system in an appropriate manner depending upon what hardware configuration you have. A Disc system will expect some program on disc to initialise, whereas a tape system will not bother with this; thus the SV's can be configured for Tape, CP/M or their own disc operating system (DOS). The Interpreter is more or less the "real" basic program. When active, it checks input via the keyboard or other sources, decides if important and if not either ignores it or generates an error message. It "drives" the basic and is partly why interpreted (as opposed to compiled) basic appears so slow.

The editor is necessary if the user is to modify or enter any sort of basic program, this is really a subsection of the interpreter but changes the subject code. Each statement code is merely the object code for a particular statement, after that statement has been read, syntax checked and variables assigned to proper buffers or locations. This is the "meat" of the basic, whereas the interpreter is the "bones", so to speak. The functions are sections of code used by more than one statement module, some are used to convert numbers, check status et cetera.

Each computer needs to relate to external devices in order to communicate with the user or provide hard copy of information. Such devices include the keyboard, screen, tape drive, disc drive, printers et cetera. In order to do so it needs a hardware configuration backed by software programs called "drivers". Drivers are normally written with particular devices in mind but a good system will be constructed so that the user can write their own drivers. I've listed the Keyboard Driver seperately since it is important for most input from the user and is the most basic peripheral for any computer.

Each computer needs a certain ammount of RAM to hold variables, pointers et cetera; these I have divided into three types: Vectors, Buffers & Tables. A vector is a location in RAM that the program will jump to and be further directed to another location. These can be

SPECTRAVIDEO

COMPUTER USERS GROUP OF TASMANIA

manipulated so that program flow can be diverted other than to the normal point of flow. Used carefully, they can add extra basic commands, perform checks or alter information flow. There is a series of vectors in high memory on the SV's, mostly used for Disc Basic.

Buffers are sections of memory used to hold temporary values for processing. The most common uses for buffers are string manipulation, file input/output and screen handling. There are several buffers in high memory, (many of which I have yet to determine the exact purpose of) used for all of these. A table is a more regular area for saving values and variables. Several tables exist both in RAM and ROM and basic maintains tables to indicate the type and location of variables, the variables themselves and for arrays. Apart from the stack, these normally produce the biggest overhead in many programs.

To work in any capacity using machine code or assembly language on a computer, the programmer should have a working knowledge of the input/output drivers, or at least the hardware connections involved, otherwise how does one manipulate the thing? Next month I shall talk about the video display and how it's interfaced with the SV and also of a few tricks you may not know.

You may remember listing 2 that was printed last month and the strange results when line 160 was listed after being altered. This is because the code used to list the program line is different from that used to edit the line. The editor normally eliminates bytes in the line less than 32 except certain functions, literals and end of line byte. The lister doesn't expect to find such characters except as part of these and so produces uncertain results. What are the bytes that do this? A list is given below:

Byte ~~~~	Use ~~~~
9	Produces a tab when listed
10	Produces a linefeed, unknown results
11	Prefixes an integer value as represented by "&O"
12	Prefixes an integer value as represented by "&H"
13	Line Number reference prefix *
14	Prefixes an integer value: Two bytes follow
15	Prefixes a single byte value: One byte follows
17--27	Equals an absolute value of byte - 17
28	Prefixes an integer value: Two bytes follow #
29	Prefixes single precision: Four bytes follow #
30	Line number reference prefix # *
31	Prefixes double precision: Eight bytes follow #

* The next two numbers indicate relative position of line in relation to current line.

Editor uses these bytes in preference to others.

The interpreter uses the prefixes to determine what to expect next, if they are used as cursor control characters in literal strings, the video driver is in control and no problem results. When edited however, the line is first listed and the editor takes the

SPECTRAVIDEO

COMPUTER USERS GROUP OF TASMANIA

result as verbatim and so the values are changed. Notice there is no "&B" prefix, this is because it is represented by ASCII as 26/42 and the "1" & "0" in the value as 31/30.

A few points aside from the above discussion; what is the least memory using method of reading data statements for sprites & machine codes? In each case each bit of data is a single byte. If these are read individually, they eat up one byte per piece of data for commas and look tacky. Listing 1 demonstrates a method that uses an optimum number of bytes. It's based on the CALL CHAR statement from the TI99/4A.

One of the annoying things about the user's manual is its general inaccuracy. Error messages for WHILE and WEND statements are listed, but these statements are not implemented. WHILE/WEND works as follows: The interpreter comes to the WHILE statement which also has a condition attached to it, if this condition is true the logic falls through to the next statement, it then continues on until it reaches the WEND statement and loops back to the WHILE statement; if the condition is false it skips to the next statement after the WEND.

A similar set of statements are the DO-REPEAT/UNTIL pair. Here the logic falls through to the UNTIL statement. If the attached condition is true it loops back to the DO/REPEAT statement; if not true it falls through to the next statement. The difference between the WHILE/WEND and REPEAT/UNTIL sets is that the REPEAT/UNTIL always executes at least once, whereas WHILE/WEND might not. Listing 2 illustrates how to simulate both sets.

Listing 1

```
10 REM *** Demonstrates Data Reading Technique ***
20 REM *** Delete REMs in lines 100 or 110 ***
25 SCREEN 1,3
30 FOR X=1 TO 3 : READ D$ : GOSUB 100 : NEXT X
60 STOP
100 REM A$="" : FOR Y=1 TO LEN(D$) STEP 2 : A$=A$+CHR$(VAL
("&H"+MID$(D$,Y,2))) : NEXT Y : SPRITE$(X)=A$ : PUT SPRITE
X,(X*16,X*16),2,X
105 REM *** Line 100 = Sprite read ***
110 REM FOR Y=1 TO LEN(D$) STEP 2 : V=VAL("&H"+MID$(D$,Y,
2)) : POKE P,V : P=P+1 : NEXT Y
115 REM *** Line 110 = Machine code read ***
116 REM *** P = Memory location for POKE ***
120 RETURN
130 DATA 003C7FDFFF3F1212,1030F1323E1E1236,FFFFFFFFFFFFFFFF
```

Listing 2

```
10 REM *** Lines 20 to 50 demonstrate WHILE/WEND ***
20 FOR X=0 TO 1 STEP 0
30 IF NOT(<Condition>) THEN X=2 : GOTO 50
40 REM *** Code to be executed goes here ***
50 NEXT X
60 REM *** Lines 70 to 100 demonstrate REPEAT/UNTIL ***
70 FOR X=0 TO 1 STEP 0
80 REM *** Code to be executed goes here ***
90 IF NOT(<Condition>) THEN X=2
100 NEXT
```

SPECTRAVIDEO

COMPUTER USERS GROUP OF TASMANIA

PROGRAM OF THE MONTH "ROBONS"

by B. Scott

Your Helicopter has crashed on an island with two swamps and inhabited by ROBONS who do not like strangers and will try to hunt them down. ROBONS always hunt in the most direct line. If they go into a swamp they drown. You can hide in the swamp for a short time but, if you stay too long, you will die from a terrible and swift disease. If you hide in a swamp and return to it or go into the other swamp your chances of catching this disease increases. Beware ROBONS may be invisible for short periods. You move with the cursor controls or joystick.

Note from ED. I changed a few bits in this program to speed it up. I think it still works OK as I have played it for a few hours and found no problems.

```
10 DEFINT A-Z:SCREEN,0
20 ON SPRITE GOSUB 750
30 FOR X=1 TO 8:READ X(X),Y(X):NEXT
40 DATA 0,-2,2,-2,2,0,2,2,0,2,-2,2,-2,0
50 DATA -2,-2
60 CLS: LOCATE14,5: PRINT" THE ROBONS": LOCATE 6,8: PRINT"PRESS 'I' FOR INFORMAT
ION                                     PRESS 'S' TO START AT ON
CE"
70 A$ = INKEY$: IF A$ = "" THEN 70
80 IF A$ = "I" THEN GOSUB 830
90 COLOR 15,2,5: SCREEN 1: CLICK OFF
100 CIRCLE(210,70),16,4,,2.6: PAINT(210,70),4
110 CIRCLE(80,130),20,4: PAINT(80,130),4
120 M = 128: E = 96: R = 20: O = 10: R1 = 240: O1 = 10: R2 = 240: O2 = 180: R3 =
20: O3 = 180: C = 1: C1 = 8: C2 = 9: C3 = 6
130 FOR T = 1 TO 8: READ A$: M$ = M$ + CHR$(VAL("&H"+A$)): NEXT: SPRITE$(0) = M$
140 FOR T = 1 TO 8: READ B: R$ = R$ + CHR$(B): NEXT: SPRITE$(1) = R$
150 RESTORE 510: FOR T = 1 TO 8: READ B: U$ = U$ + CHR$(B): NEXT: SPRITE$(2) = U
$
160 RESTORE 510: FOR T = 1 TO 8: READ B: V$ = V$ + CHR$(B): NEXT: SPRITE$(3) = V
$
170 RESTORE 510: FOR T = 1 TO 8: READ B: W$ = W$ + CHR$(B): NEXT: SPRITE$(4) = W
$
180 SPRITE ON
190 M = M+I: E = E+J
200 IF STICK(0)+STICK(1)=0 GOTO 230
210 I=X(STICK(0)+STICK(1))
220 J=Y(STICK(0)+STICK(1))
230 PUT SPRITE 0, (M,E),13
240 PUT SPRITE 1, (R,O),C
```

SPECTRAVIDEO

COMPUTER USERS GROUP OF TASMANIA

```
250 PUT SPRITE 2, (R1,01),C1
260 PUT SPRITE 3, (R2,02),C2
270 PUT SPRITE 4, (R3,03),C3
280 R=R+SGN(M-R)
290 O=O+SGN(E-O)
300 R1=R1+SGN(M-R1)
310 O1=O1+SGN(E-O1)
320 R2=R2+SGN(M-R2)
330 O2=O2+SGN(E-O2)
340 R3=R3+SGN(M-R3)
350 O3=O3+SGN(E-O3)
360 IF M > 243 THEN M = 243: I=0 ELSE IF M < 12 THEN M = 12: I = 0
370 IF E < 10 THEN E = 10: J = 0 ELSE IF E > 182 THEN E = 182: J = 0
380 IF POINT(R,O) = 4 THEN GOSUB 520
390 IF POINT(R1,O1) = 4 THEN GOSUB 570
400 IF POINT(R2,O2) = 4 THEN GOSUB 620
410 IF POINT(R3,O3) = 4 THEN GOSUB 670
420 IF POINT (M,E) <> 2 THEN GOSUB 720
430 IF LU > 150 THEN 740
440 IF NU < 6 THEN 180
450 COLOR 15,4: SCREEN 0
460 PLAY"V12 T255 O4 L8 C F R8 C F R8 C F R4 R8 C F G F E R8 F G R4 R8 C E R8 C
E R8 C E R4 R8 C E F E D R8 E F R4 R8"
470 PRINT"WELL DONE YOU OUTLASTED THEM HELP IS ON THE WAY."
480 PLAY"V12;T255;O6;L8CFR8CFR8CFR4R8CFGR8FGR4R8CER8CER8CER4R8CEFEDR8EFR4R8"
490 CLICK ON: END
500 DATA 1C,1C,88,7E,1D,1D,14,36
510 DATA 60,126,195,195,255,255,195,195
520 C = 0: PUT SPRITE 1, (R,O),C:
530 NU = NU+1: R = 20: O = 10:
540 FOR T = 0 TO 5: SOUND 0,255 - 50*T: SOUND 8,15 - 3*T: FOR H = 1 TO 25: NEXT
H,T
550 C = 1
560 RETURN
570 C1 = 0: PUT SPRITE 2, (R1,O1),C1
580 R1 = 240: O1 = 10:
590 FOR T = 0 TO 5: SOUND 0,50*T: SOUND 8,15 - 3*T: FOR H = 1 TO 25: NEXT H,T
600 C1 = 8
610 RETURN
620 C2 = 0: PUT SPRITE 3, (R2,O2),C2
630 R2 = 240: O2 = 180:
640 FOR T = 0 TO 6: SOUND 2,255-40*T: SOUND 9,15 - (5*T)/2: FOR H = 1 TO 20: NEX
T: SOUND 9,0: NEXT
650 C2 = 9
660 RETURN
670 C3 = 0: PUT SPRITE 4, (R3,O3),C3
680 R3 = 20: O3 = 180:
690 FOR T = 0 TO 6: SOUND 2,40*T: SOUND 9,15 - (5*T)/2: FOR H = 1 TO 20: NEXT: S
OUND 9,0: NEXT
700 C3 = 6
710 RETURN
720 LU = LU+1: SOUND 4,255-(3*LU)/2: SOUND 10,8: FOR H = 1 TO 20: NEXT: SOUND10,
0
730 RETURN
740 COLOR 15,1: SCREEN 0: LOCATE 3,10: PRINT"YOU HAVE JUST DIED OF THE DREADED
LURGY.": CLICK ON: END
750 SPRITE OFF
```


SPECTRAVIDEO

COMPUTER USERS GROUP OF TASMANIA

```
760 IF M=R AND E=0 AND C<>0 THEN 810
770 IF M=R1 AND E=01 AND C1<>0 THEN 810
780 IF M=R2 AND E=02 AND C2<>0 THEN 810
790 IF M=R3 AND E=03 AND C3<>0 THEN 810
800 RETURN
810 PLAY"v 8 04 L 4 D R64 D 03 A 04 L2 D L 4 03 A"
820 COLOR 13,1 :SCREEN 0: LOCATE 0,10: PRINT "THE ROBONS HAVE CAUGHT YOU!! YOU
HAVE TO WORK IN THEIR SAND MINES. GOODBYE!": CLICK ON: END
830 CLS: PRINT"Your helicopter has crashed on an island with two swamps and
inhabited byROBONS who do not like strangers and will try to hunt them down.
ROBONS always hunt in the most direct line."
840 PRINT"If they go into a swamp they drown."
850 PRINT:PRINT"You can hide in a swamp for a short time but, if you stay too
long, you will die from a terrible and very swift disease. If you hide in a
swamp and return to it or go into the other"
860 PRINT"your chance of catching this disease increases."
870 PRINT: PRINT"Beware ROBONS may be invisible for short periods."
880 PRINT: PRINT"CONTROL YOUR MOVEMENTS WITH THE CURSOR CONTROLS OR JOYSTICK."
890 PRINT" Press any key"
900 A$ = INKEY$: IF A$ = "" THEN 900
910 RETURN
```

Note from ED.

The text in lines 830-890 looks a bit messy. This is because the program has been printed on an 80 column printer. It will appear OK on the normal SPECTRAVIDEO 40 column screen.



SPECTRAVIDEO

COMPUTER USERS GROUP OF TASMANIA

JESSUPS

SV-318, Data Cassette and 4 Programs	\$ 395
SV-328, Data Cassette and 4 Programs	\$ 699
SV-328, Expander, Disk, Control Card & CP/M	\$1390
Coleco Adapter	\$ 119
Quick Shot III Joystick	\$ 35

For other prices on SPECTRAVIDEO equipment call
Andrew Ramage (003 316933).

The following was seen on an SV Computer

ACHTUNG

=====

ALLES LOOKENPEEPER

DIES MACHINE IS NICHT FUR
GEFINGERPOKEN MITTEN GRABEN.

IS EASY SCHNAPPEN DER SPRINGENWORK,
BLOWEN UND POPPENCORKEN MIT SPITEN
UND SPARKEN. IS NICHT FUR TOUCHEN
BY DAS DUMMKOFFEN UND RUBBER NECKEN
SIGHTSEERS.

KEEP EN HANDS IN DAS POCKETS,
RELAXEN, UND WATCH DAS BLINKEN
LIGHTS.